

# NAG Library Routine Document

## F07TJF (DTRTRI)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07TJF (DTRTRI) computes the inverse of a real triangular matrix.

### 2 Specification

```
SUBROUTINE F07TJF (UPLO, DIAG, N, A, LDA, INFO)
  INTEGER          N, LDA, INFO
  REAL (KIND=nag_wp) A(LDA,*)
  CHARACTER(1)    UPLO, DIAG
```

The routine may be called by its LAPACK name *dtrtri*.

### 3 Description

F07TJF (DTRTRI) forms the inverse of a real triangular matrix  $A$ . Note that the inverse of an upper (lower) triangular matrix is also upper (lower) triangular.

### 4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

### 5 Arguments

- 1: UPLO – CHARACTER(1) *Input*  
*On entry:* specifies whether  $A$  is upper or lower triangular.  
 UPLO = 'U'  
      $A$  is upper triangular.  
 UPLO = 'L'  
      $A$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: DIAG – CHARACTER(1) *Input*  
*On entry:* indicates whether  $A$  is a nonunit or unit triangular matrix.  
 DIAG = 'N'  
      $A$  is a nonunit triangular matrix.  
 DIAG = 'U'  
      $A$  is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.  
*Constraint:* DIAG = 'N' or 'U'.
- 3: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .

4: A(LDA,\*) – REAL (KIND=nag\_wp) array Input/Output

**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .

*On entry:* the  $n$  by  $n$  triangular matrix  $A$ .

If UPLO = 'U',  $A$  is upper triangular and the elements of the array below the diagonal are not referenced.

If UPLO = 'L',  $A$  is lower triangular and the elements of the array above the diagonal are not referenced.

If DIAG = 'U', the diagonal elements of  $A$  are assumed to be 1, and are not referenced.

*On exit:*  $A$  is overwritten by  $A^{-1}$ , using the same storage format as described above.

5: LDA – INTEGER Input

*On entry:* the first dimension of the array A as declared in the (sub)program from which F07TJF (DTRTRI) is called.

*Constraint:*  $LDA \geq \max(1, N)$ .

6: INFO – INTEGER Output

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element  $\langle value \rangle$  of the diagonal is exactly zero.  $A$  is singular its inverse cannot be computed.

## 7 Accuracy

The computed inverse  $X$  satisfies

$$|XA - I| \leq c(n)\epsilon|X||A|,$$

where  $c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

Note that a similar bound for  $|AX - I|$  cannot be guaranteed, although it is almost always satisfied.

The computed inverse satisfies the forward error bound

$$|X - A^{-1}| \leq c(n)\epsilon|A^{-1}||A||X|.$$

See Du Croz and Higham (1992).

## 8 Parallelism and Performance

F07TJF (DTRTRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately  $\frac{1}{3}n^3$ .

The complex analogue of this routine is F07TWF (ZTRTRI).

## 10 Example

This example computes the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} 4.30 & 0.00 & 0.00 & 0.00 \\ -3.96 & -4.87 & 0.00 & 0.00 \\ 0.40 & 0.31 & -8.02 & 0.00 \\ -0.27 & 0.07 & -5.95 & 0.12 \end{pmatrix}.$$

### 10.1 Program Text

```

Program f07tjfe

!      F07TJF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dtrtri, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter    :: diag = 'N'
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, lda, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:, :)
!      .. Executable Statements ..
      Write (nout,*) 'F07TJF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
         Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
         Read (nin,*)(a(i,1:i),i=1,n)
      End If

!      Compute inverse of A
!      The NAG name equivalent of dtrtri is f07tjf
      Call dtrtri(uplo,diag,n,a,lda,info)

!      Print inverse

      Write (nout,*)
      Flush (nout)

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf(uplo,diag,n,n,a,lda,'Inverse',ifail)

End Program f07tjfe

```

## 10.2 Program Data

F07TJF Example Program Data

```
4                               :Value of N
'L'                             :Value of UPLO
4.30
-3.96  -4.87
0.40   0.31  -8.02
-0.27  0.07  -5.95  0.12  :End of matrix A
```

## 10.3 Program Results

F07TJF Example Program Results

```
Inverse
      1          2          3          4
1      0.2326
2     -0.1891    -0.2053
3      0.0043    -0.0079   -0.1247
4      0.8463    -0.2738   -6.1825    8.3333
```

---